Curso: 2006-2007

INGENIERÍA TÉCNICA INDUSTRIAL - Esp. Mecánica

FUNDAMENTOS MATEMÁTICOS DE LA INGENIERÍA

Práctica nº 8: Ecuaciones diferenciales ordinarias

RESUMEN: Estudiamos en esta práctica el cálculo con MAPLE de las soluciones de ecuaciones diferenciales ordinarias, así como la representación gráfica de dichas soluciones.

COMANDOS MÁS IMPORTANTES:

diff( ), dsolve( ), assign( ), unapply( ), dsolve(,numeric), odeplot( )

Ecuación diferencial ordinaria

Una **ecuación diferencial ordinaria (EDO)** es una ecuación que contiene las derivadas de una función desconocida *y(x)* respecto a una variable independiente $x \in [a,b]$:

$$F(x, y, \frac{dy}{dx}, \frac{d^2y}{dx^2}, \ldots, \frac{d^ny}{dx^n}) = 0$$

Se llama **orden de la ecuación diferencial** al orden de la derivada de mayor orden que aparece en dicha ecuación.

Una función y(x) suficientemente derivable (a,b) que verifica dicha ecuación diferencial se denomina **solución de la ecuación diferencial**.
Una ecuación diferencial puede tener soluciones o no y cuando la tiene no suele ser única. Si se puede determinar, el conjunto de todas las soluciones posibles de una ecuación diferencial se llama **solución general de la ecuación diferencial.**

MAPLE puede calcular soluciones de muchas ecuaciones diferenciales. Para ello primero tenemos que definir dicha ecuación. Para expresar las derivadas que aparecen en la ecuación tendremos que usar el comando

diff(y(x),x$n)

donde **y(x)** es la función incógnita, **x** es la variable independiente y **n** es el orden de derivación. Si n=1 se puede poner simplemente **diff(y(x),x)**.

Ejemplo: Definir la ecuación diferencial $x^2\frac{d^2y}{dx^2} - 2x\frac{dy}{dx} + 2y = 0$

```
> ec:=x^2*diff(y(x),x$2)-2*x*diff(y(x),x)+2*y(x)=0;
```

Para resolver ecuaciones diferenciales existen varias formas. La más general es usando el comando

dsolve(ecuacion,y(x))

donde **ecuacion** es la ecuación diferencial que tiene como incógnita **y(x)**.

Ejemplo: Resolver la ecuación definida en el ejemplo anterior.

```
> s:=dsolve(ec,y(x));
```

En este ejemplo aparecen dos constantes de integración _C1 y _C2 arbitrarias, pues MAPLE nos ha dado todas las posibles soluciones de esta ecuación, esto es, su solución general.

En muchos problemas no necesitaremos calcular todas las soluciones posibles de una ecuación, sino encontrar aquella o aquellas que cumplan ciertas condiciones iniciales. Esto es lo que denomina **problema de Cauchy.** Con dsolve también podemos abordar dichos problemas, incluyendo los valores iniciales junto con la ecuación:

dsolve({ecuacion,condiciones},y(x))

donde las **condiciones** en el punto inicial $x = x0$ en las derivadas hay que escribirlas con el comando **D(y)(x0), (D@@2)(y)(x0)**, etc.

Ejemplo: Resolver los problemas de Cauchy:

(a) $\dfrac{d^2y}{dx^2} - 2\dfrac{dy}{dx} + 4y = x,\ y(0) = 1, y'(0) = 0$ (b) $\dfrac{dy}{dt} + \sqrt{1+t^2}\,y = 0,\ y(0) = \sqrt{5}$ (Problema 5d)

```
> ec1:=diff(y(x),x$2)-2*diff(y(x),x)+4*y(x)=x;
> s1:=dsolve({ec1,y(0)=1,D(y)(0)=0},y(x));
> ec2:=diff(y(t),t)+sqrt(1+t^2)*y(t)=0;
> s2:=dsolve({ec2,y(0)=sqrt(5)},y(t));
```

Otra posibilidad que nos da MAPLE es dibujar estas soluciones. En este último ejemplo vemos que en la variable s2 está almacenada la solución particular del problema de Cauchy en forma de ecuación. Para poderla dibujar tenemos que tener extraer de s2 dicha solución almacenándola como una función. Esto se hace con el comando

assign(var)

junto con la orden **unapply(expr,var)** que vimos en la práctica 6. Veámoslo con el ejemplo:

```
> assign(s2);
```

Así lo que hemos hecho es crear una variable denotada por y(t) que almacena la solución:

```
> y(t);
```

Pero aún no es una función. Por ejemplo, si queremos calcular su valor en t=5 no nos sale:

```
> y(5);
```

Para que sea una función, usamos unapply:

```
> y:=unapply(y(t),t);
> y(5);
```

Y esta función se puede dibujar con **plot:**

```
> plot(y(t),t=0..10,title=`Solucion del problema de Cauchy`);
```

Hay muchos problemas de Cauchy que MAPLE no puede resolver explícitamente. Por ejemplo, el problema

$$(x+1)y' = 8e^{x^2},\ y(0)=1$$

```
> restart:
> ec:=diff(y(x),x)*(x+1)=8*exp(x^2);
> s:=dsolve({ec,y(0)=1},y(x));
```

En estos casos, MAPLE nos puede dar una aproximación a dicha solución usando la opción

dsolve({ecuacion,condiciones},y(x),numeric)

```
> q:=dsolve({ec,y(0)=0},y(x),numeric);
```

Este comando nos devuelve una función q(x) que podemos usar para calcular valores concretos de la solución, como por ejemplo

```
> q(2);
```

También podemos dibujar la gráfica de la solución, pero en este caso no funciona la orden plot. Para ello tenemos que usar el comando

odeplot(q,[x,y(x)],a..b)

del paquete **plots.** Aquí **q** es la función que creamos con el dsolve(...,numeric), **[x,y(x)]** son las variables y **a..b** es el rango donde va a variar la independiente.

```
> with(plots):
> odeplot(q,[x,y(x)],0..2);
```

Ejemplo: (Problema 24) La semivida del cobalto radiactivo es de 5.27 años. Supóngase que un accidente nuclear ha provocado que el nivel de este cobalto ascienda en una región hasta 100 veces el nivel aceptable para la vida humana. ¿Cuánto tiempo pasará hasta que la región vuelva a ser habitable?

Solución

La semivida de una sustancia radiactiva es el tiempo que tarda en que una cantidad disminuya a la mitad. Si llamamos q(t) a la cantidad de cobalto que hay en el tiempo t, tenemos que q(5.27)=q(0)/2. Denotando Q= cantidad de cobalto aceptable para la vida humana, tenemos que q(0)=100Q. Lo que nos pide el problema es hallar el tiempo T de forma que q(T)=Q.
Por otra parte, sabemos que la velocidad a la que se desintegra una sustancia radiactiva es proporcional a la cantidad presente de dicha sustancia, esto es:

$$\frac{dq(t)}{dt} = -rq(t)$$

donde r >0 es la razón de decaimiento de la sustancia. Por tanto tenemos que resolver primero esta ecuación diferencial con la condición inicial dada:

```
> restart:
> ec:=diff(q(t),t)=-r*q(t);
> sol:=dsolve({ec,q(0)=100*Q},q(t));
> assign(sol);
> q(t);
> q:=unapply(q(t),t);
> q(5.27);
```

Necesitamos saber quién es r, que depende del tipo de sustancia que tengamos. Para ello usamos la semivida del cobalto:

```
> solr:=solve(q(5.27)=50*Q,r);
> r:=solr;
```

Finalmente sólo nos queda saber cuánto tiempo pasará hasta que q(T)=Q:

```
> solT:=solve(q(T)=Q,T);
> T:=solT;
```

Así que se necesitarán más de 35 años para que sea habitable la zona. Esto lo podemos ver gráficamente. Si suponemos que la cantidad aceptable para la vida es Q=1, se tiene

```
> Q:=1;
> plot(q(t),t=0..T);
```

Ejemplo: (Problema 28) Una bala se introduce en una tabla de h=10 cm de espesor con una velocidad v0=200 m/seg traspasándola con la velocidad v1=80 m/seg. Suponiendo que la resistencia de la tabla al movimiento de la bala es proporcional al cuadrado de la velocidad, hallar el tiempo que tarda la bala en atravesar la tabla.

Solución

La fuerza que se ejerce sobre la bala dentro de la tabla es la resistencia de la tabla, que nos dicen que es proporcional al cuadrado de su velocidad. Si llamamos m a la masa de la bala, v(t) a su velocidad en el instante t y k>0 a la constante de proporcionalidad de la resistencia de la tabla, usando la segunda ley de Newton (F=m a), tenemos que se verifica la ecuación:

$$m\frac{dv(t)}{dt}=-kv^2(t)$$

con la condición inicial v(0)=200. La podemos resolver con MAPLE:

```
> restart:
> ec:=m*diff(v(t),t)=-k*(v(t))^2;
> sol:=dsolve({ec,v(0)=200},v(t));
> assign(sol);
> v:=unapply(v(t),t);
```

Si llamamos T al tiempo en que la bala sale de la tabla, por hipótesis v(T)=80. Así que para calcular T tenemos que resolver esta ecuación respecto de T:

```
> solT:=solve(v(T)=80,T);
> T:=solT;
```

Pero para calcular T exactamente necesitamos saber quién es m/k. Para ello es para lo que usaremos el dato de que la tabla mide 10 cm de grueso. Sabemos que

$$v(t)=\frac{dx(t)}{dt}$$

donde x(t) denota a la posición de la bala en el tiempo t. Así que

$$x(t)=\int v(t)dt$$

Para crear la función x(t) como la integral de v(t) usamos el comando unapply que vimos para almacenar una expresión como una función de t

```
> x:=unapply(int(v(t),t)+C,t);
```

Por hipótesis, tenemos que x(0)=0, así que la constante de integración sale de resolver esta ecuación:

```
> solC:=solve(x(0)=0,C);
```

```
> C:=solC;
> simplify(x(t));
```
Y tiene que cumplirse que x(T)=10:
```
> ecu:=simplify(x(T))=10;
```
Ya de aquí vemos de forma sencilla que m/k=10/(ln 5- ln 2), pero para resolverlo mejor:
```
> solfin:=solve(ecu,k);
> k:=solfin;
```
Y la solución es
```
> Tfin:=simplify(T);
> evalf(Tfin);
```
Si la bala tiene una masa de m=1 gr., podemos ver gráficamente cómo varía la velocidad y su posición a lo largo del tiempo:
```
> m:=1;
> plot(v(t),t=0..Tfin);
> plot(x(t),t=0..Tfin);
```